
CMOS OpAmp Course Project 

 Section 1: 
 

Architecture : Telescopic two stage Op-Amp 

 

 Full schematic

 



 

Core OpAmp  

First stage: Telescopic  

2nd Stage : Common Source stage 

 

 

 

 



 

CMFB 

 
 

 

 

 

 

 

 



 

Bias Generator 

 
 

 



 

Section 2: 
 

Abstract Design Flow 

 

1> According to the design specification, telescopic opamp is used 

in the first stage and common source configuration is used in 

the second stage. 

2> From Slew rate specification and by deciding compensation 

capacitor, the bias current requirement is fixed to ensure 

sufficient slewing period. The bias current of 2nd stage is fixed 

nearly 5 times the first stage so that during slewing it does not 

load first stage and we can get better slew rate. 

3>    Gm1 can be found from wu from which input transistor 

overdrive can be decided. 

4> From the input common mode specifications, we can decide 

overdrive voltages and over drive voltages of other upper 

MOSFETS 

5>  Once current and overdrive is known, biasing can be decided. 

6> We need to ensure that C1<<Cc and C1<<Cl and for this we 

keep the W/L ratios minimum for transistors connected to the 

output of stage 1. 

7> The dominant pole must be kept as close to origin so as to get 

wu of desired specification. This can be achieved by frequency 

compensation technique. 

8> The Rz1,2 and Cc1,2 are used for frequency compensation but 

their introduction makes the system third order and may cause 

instability due to insufficient phase margin introduced due to 

third pole. 



9> Therefore , pole 2 and zero cancellation is necessary to 

improve unity gain bandwidth as well as phase margin to 

desired specifications. 

10> Rz1,2 must be high enough so that zero introduced by Cc1,2 

is on the left half of s-plane so that stability can be ensured. 

11> Pole 3 must be far away from Wu. 

12> The CMFB should be designed with reasonably high gain to 

ensure less errors and make the output stage stable. Its open 

loop phase margin should be high enough to ensure stablilty at 

high frequency. 

13> Reference  generators are designed for stable voltage 

biasing to inputs of different MOSFETs. Current mirroring 

technique is used by chossig appropriate W/L ratios for the tail 

current source of 1st , 2nd stage and tail current and reference 

source of CMFB.  

 

 

 

 

 

 

 

 

 

 

 



 Section 3:  

 

Detailed Design flow of OpAmp  

 The slew rate is given as:  

Slew Rate  ≥  20 V/us 

First Stage Bias Current 

 S.R. = 
𝑰𝒃𝒊𝒂𝒔 𝟏

𝑪𝒄
  

 

 CL  =5pF load capacitance is given and we know that Cc≥ 0.3 *CL 

So, choosing  Cc=2.5pF 

 Therefore ,  Ibias ≥   50 uA  

 Choosing Ibias = 114uA for the tail current source of the first stage to 

sufficient slew rate. 

Second Stage Bias Current 

 S.R. = 
𝑰𝒃𝒊𝒂𝒔 𝟐

𝑪𝒄 +𝑪𝑳
 

 Ibias2 ≥ 150uA 

 Choosing Ibias2 = 572 uA  

 

 

 

 

 

 

 



W and L Values of Transistors  

 

 

 

1>  1st  Stage OpAmp 

 

Name of transistors W and L values 

Ms W=33.33um, L=1um 

M1,2 W=50um, L=1um 

M3,4 W=20.65um, L=1um 

M5,6 W=33.6um, L=1um 

M7,8 W=55.38um, L=1um 

 

2nd  Stage OpAmp 

Name of transistors W and L values 

M9,11 W=20.1um, L=0.5um 

M10,12 W=81um, L=0.3um 

 

2>  Common Mode Feedback (CMFB) 

        

Name of transistors W and L values 

Ms2 W=15um, L=1um 

M14,15 W=15um, L=1um 

M16,17 W=25um, L=1um 

Ms3 W=15um, L=1um 

M18,19 W=25um, L=1um 

 

 



3> Reference Generator 

Name of transistors W and L values 

Mp1 W=2um, L=1um 

Mp2 W=2um, L=1um 

Mp3 W=8.7um, L=1um 

Mp4 W=33.7um, L=1um 

Mp5 W=2um, L=1um 

Mp6 W=3.63um, L=1um 

Mp7 W=2um, L=1um 

Mp8 W=2um, L=1um 

Mp9 W=5.1um, L=1um 

Mp10 W=2um, L=1um 

Mp11 W=8.5um, L=1um 

Mp12 W=11.83um, L=1um 

Mp13 W=2um, L=1um 

 

 

 

Netlist of entire OpAmp Design: 
 

********************** First stage design ********************** 

 

.include MOS.txt 

Ms1 1 gp9 0 0 CMOSN L=1um W=33.33um as=2.88p ad=2.88p ps=16.72u pd=16.72u 

 

M1 2 g1 1 1 CMOSN L=1um W=50um  as=10.8p ad=10.8p ps=780u pd=780u 

M2 3 g2 1 1 CMOSN L=1um W=50um  as=10.8p ad=10.8p ps=780u pd=780u 

M3 4 gp5 2 2 CMOSN L=1um W=20.65um as=4.4604p ad=4.4604p ps=25.5u pd=25.5u 

M4 5 gp5 3 3 CMOSN L=1um W=20.65um  as=4.4604p ad=4.4604p ps=25.5u pd=25.5u 

 

M5 4 gp3 6 6 CMOSP L=1um W=33.6um  as=7.2576p ad=7.2576p ps=41.04u pd=41.04u 

M6 5 gp3 7 7 CMOSP L=1um W=33.6um  as=7.2576p ad=7.2576p ps=41.04u pd=41.04u 

M7 6 g7 8 8 CMOSP L=1um W=55.38um  as=7.2576p ad=7.2576p ps=67.88u pd=67.88u 

M8 7 g7 8 8 CMOSP L=1um W=55.38um  as=7.2576p ad=7.2576p ps=67.88u pd=67.88u 

 

*RG1 4 n 1000G 

*RG2 5 n 1000G 



 

Vdd 8 0 dc 1.8 

 

*Vbs1 gs1 0 dc 0.58 

*Vb1 g1 0 dc 0.93 ac 1v 

*Vb2 g2 0 dc 0.93 

*Vb3 g3 0 dc 1.52 

*Vb4 g4 0 dc 1.52 

 

*Vb5 g5 0 dc 0.8 

*Vb6 g6 0 dc 0.8 

*Vb7 g7 0 dc 1.181 

*Vb8 g8 0 dc 1.181 

 

*********************** Second Stage Design ********************** 

 

M9 9 gp7 0 0 CMOSN L=0.5um W=20.1um as=4.068p ad=4.068p ps=23.32u pd=23.32u 

M10 9 5 8 8 CMOSP L=0.3um W=81um as=29.216p ad=29.216p ps=162.72u pd=162.72u 

*Vb9 g9 0 dc 0.78 

 

M11 10 gp7 0 0 CMOSN L=0.5um W=20.1um as=4.068p ad=4.068p ps=23.32u pd=23.32u 

M12 10 4 8 8 CMOSP L=0.3um W=81um   as=29.232p ad=29.232p ps=162.72u pd=162.72u 

*Vb11  g11 0 dc 0.78 

 

Cl 9 10 5pF 

cc1 x2 10 2.5pf 

cc2 x1 9 2.5pf 

Rz1 4 x2 1.9k 

Rz2 5 x1 1.9k 

 

************************** CMFB design ***************************  

 

MS2 11 gp11 0 0 CMOSN L=1um W=15um  

M14 12 4 11 11 CMOSN L=1um W=15um  

M15 g7 gp13 11 11 CMOSN L=1um W=15um  

M16 12 12 8 8 CMOSP L=1um W=25um  

M17 g7 12 8 8 CMOSP L=1um W=25um  

 

*Vbs2 gs2 0 dc 0.54 

*Vg14 g14 0 dc 1.101 

 

*Vref ref 0 dc 1.103449 

 

MS3 13 gp11 0 0 CMOSN L=1um W=15um  

M18 12 5 13 13 CMOSN L=1um W=15um  

M19 g7 gp13 13 13 CMOSN L=1um W=15um  

*M20 14 14 8 8 CMOSP L=1um W=25um  



*M21 g8 14 8 8 CMOSP L=1um W=25um  

 

*Vbs3 gs3 0 dc 0.54 

*Vg18 g18 0 dc 1.101 

 

********************* Reference Generator for bias voltages************* 

 

mp1 gp1 gp1 0 0 cmosn l=1u w=2u 

mp2 gp3 gp1 0 0 cmosn l=1u w=2u 

mp3 gp3 gp3 8 8 cmosp l=1u w=8.7u 

rp1 8 gp1 10k 

vdd 8 0 1.8  

 

***********************MOS 3,4 bias 1.52 

 

mp4 gp5 gp3 8 8 cmosp l=1u w=33.7u 

mp5 gp5 gp5 0 0 cmosn l=1u w=2u 

 

***********************MOS 10,12 bias 0.78 

mp6 gp7 gp3 8 8 cmosp l=1u w=3.63u 

mp7 gp7 gp7 0 0 cmosn l=1u w=2u 

 

***********************Ms1 tail bias 0.58 

mp8 gp9 gp3 8 8 cmosp l=1u w=2u 

mp9 gp9 gp9 0 0 cmosn l=1u w=5.1u 

 

***********************CMFB tail ms2 ms3 bias 0.54 

mp10 gp11 gp3 8 8 cmosp l=1u w=2u 

mp11 gp11 gp11 0 0 cmosn l=1u w=8.5u 

 

***********************CMFB reference 1.103449 

mp12 gp13 gp3 8 8 cmosp l=1u w=11.83u 

mp13 gp13 gp13 0 0 cmosn l=1u w=2u 

 

***************************************************************************** 

Rs1 sr1 g1 1meg 

Rs2 sr2 g2 1meg 

Rs3 g1 9 1meg 

Rs4 g2 10 1meg 

 

*Vsr1 sr1 0 pulse(0.775 1.225 1n 1n 1n 1u 2u) 

*Vsr2 sr2 0 pulse(1.225 0.775 1n 1n 1n 1u 2u) 

Vsr1 sr1 0 sin(0.93 0.4 1k 0 0) 

Vsr2 sr2 0 sin(0.93 -0.4 1k 0 0) 

.tran 0 4m 

 

***************************************************************************** 



*.ac dec 100 100Hz 100G 

 

.control 

run 

 

plot v(sr1) v(10,9) 

*plot deriv(v(10,9)) 

 

*plot vdb(10,9) xlog 

*plot vp(10,9) xlog 

*plot vdb(4,5) 

*op 

*show 

*print v(4) v(5) v(9) v(10) 

 

.endc 

.end 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 



 Section 4:  

 

1> Design of CMFB 

 The biasing requirement of transistors M7 and M8 is 1.181V. 

 The output bias voltage at nodes 4 and 5 is 1.103V 

 Therefore, the reference voltage of CMFB is set to 1.103V which is 

made available from stable reference generator whereas the inputs to 

gate of M14 and M18 is made directly available from nodes 4 and 5. 

 Vref = 1.103V is applied to gate of 15 and M19. 

 Vg14 is made available from node 4. 

 Vg18 is made available from node 5. 

 Current of CMFB is so chosen that it is low enough to get higher 

gain so that the output of opamp is stable w.r.t to variatons in dc 

input voltage. 

o Therefore, setting  Ibias =32 uA for MS2 and MS3. 

 

 Common Mode Feedback (CMFB) W and L values.  

 Since we now know the values of current in each branch of 

CMFB and assuming overdrive values for each transistors, we obtain 

their W and L values.  

 

 ID = 0.5 * un *Cox * (W/L) * (Voverdrive)2 

       Assuming L=1um 

Name of transistors W and L values 

Ms2 W=15um, L=1um 

M14,15 W=15um, L=1um 

M16,17 W=25um, L=1um 

Ms3 W=15um, L=1um 

M18,19 W=25um, L=1um 



 

 

 

 

 Netlist of CMFB 

.include MOS.txt 

MS2 11 gs2 0 0 CMOSN L=1um W=15um  

M14 12 g14 11 11 CMOSN L=1um W=15um  

M15 g7 ref 11 11 CMOSN L=1um W=15um  

M16 12 12 8 8 CMOSP L=1um W=25um  

M17 g7 12 8 8 CMOSP L=1um W=25um  

Vbs2 gs2 0 dc 0.54  

Vg14 g14 0 dc 1.103449 ac 1 

Vdd 8 0 1.8 

Vref ref 0 dc 1.103449 

MS3 13 gs3 0 0 CMOSN L=1um W=15um  

M18 12 g18 13 13 CMOSN L=1um W=15um  

M19 g7 ref 13 13 CMOSN L=1um W=15um  

Vbs3 gs3 0 dc 0.54 

Vg18 g18 0 dc 1.103449 

.ac dec 100 100Hz 100G 

.control 

run 



plot vdb(g7) xlog 

plot vp(g7) xlog 

.end 

.end 

 

 Gain Plot 

 

 

 Gain in dB for CMFB = 34.4828 dB = 52.98  

 Unity Gain Bandwidth of CMFB = 395 MHz  

 



 

 

 

 

 

 

 Phase Plot 

 

 

Phase Margin of CMFB = 51.08 

 

 

 

 



 

 

 

 

 

 

2> Design of Reference Generator 

 

Name of transistors W and L values 

Mp1 W=2um, L=1um 

Mp2 W=2um, L=1um 

Mp3 W=8.7um, L=1um 

Mp4 W=33.7um, L=1um 

Mp5 W=2um, L=1um 

Mp6 W=3.63um, L=1um 

Mp7 W=2um, L=1um 

Mp8 W=2um, L=1um 

Mp9 W=5.1um, L=1um 

Mp10 W=2um, L=1um 

Mp11 W=8.5um, L=1um 

Mp12 W=11.83um, L=1um 

Mp13 W=2um, L=1um 

 Netlist for Reference Generator 

 

.include MOS.txt 

mp1 gp1 gp1 0 0 cmosn l=1u w=2u 

mp2 gp3 gp1 0 0 cmosn l=1u w=2u 



mp3 gp3 gp3 8 8 cmosp l=1u w=8.7u 

rp1 8 gp1 10k 

vdd 8 0 1.8  

 

***********************MOS 3,4 bias 1.52 

 

mp4 gp5 gp3 8 8 cmosp l=1u w=33.7u 

mp5 gp5 gp5 0 0 cmosn l=1u w=2u 

 

***********************MOS 10,12 bias 0.78 

mp6 gp7 gp3 8 8 cmosp l=1u w=3.63u 

mp7 gp7 gp7 0 0 cmosn l=1u w=2u 

 

***********************Ms1 tail bias 0.58 

mp8 gp9 gp3 8 8 cmosp l=1u w=2u 

mp9 gp9 gp9 0 0 cmosn l=1u w=5.1u 

 

***********************CMFB tail ms2 ms3 bias 0.54 

mp10 gp11 gp3 8 8 cmosp l=1u w=2u 

mp11 gp11 gp11 0 0 cmosn l=1u w=8.5u 

***********************CMFB reference 1.103449 

mp12 gp13 gp3 8 8 cmosp l=1u w=11.83u 

mp13 gp13 gp13 0 0 cmosn l=1u w=2u 



.control 

run 

op 

show 

print v(gp3) v(gp5) v(gp7) v(gp9) v(gp11) v(gp13)  

.endc 

.end 

 

 

The above netlist is run in ngspice and the result is shown 

above. All the voltages shown above are the bias voltages given to 

OpAmp and CMFB as specified in the netlist comments. 

The voltages from reference generator is given to the gate of 

o M5,M6 

o M3,M4 

o M10, M12 

o MS1 

o MS2, MS3 

o M15 , M19 

respectively according to above results. 

 



 

 Section 5:  

DC Simulaton results  

DC operating points from ngspice simulation for all transistors are 

shown below as follows: 

The above ngspice result shows node voltages v(4) and v(5) of the first 

stage of opamp and v(9) and v(10) for the second stage. 

 

All the voltages shown above are the bias voltages given by 

reference generator to OpAmp and CMFB. 

 



 Section 6:  

AC Simulation Results 

Netlist          

Magnitude Plot of Open loop differential voltage gain in dB 

 

The above plot highlights the location of dominant pole and the third 

pole, the second pole being compensated by a zero.  

Pole1 =500Hz 

Pole3 = 105MHz 

Magnitude of open loop differential voltage gain in dB = 98.275dB  

Unity Gain Bandwidth = 89.05 MHz 

 



 

Phase Plot: 

 

 

Phase Margin = -1.988 * 180 / 𝝅 + 180  =  66.09 degrees  

 

 

 

 

 

 

 



 

Magnitude Plot of  Open loop common-mode voltage gain in dB, as 

a function of frequency 

 

 

 

 

 

 

 

 

 

 



 

 Section 7:  

Simulation results of slew rate and step response to +450 mV 
and -450 mV differential input step voltage in closed loop mode with 

voltage gain of -1 
 

 

 



 

 



 

From above graph, Slew Rate is = 68V/uS 

Slew Rate 
 

Partial netlist for slew rate 
Rs1 sr1 g1 1meg 
Rs2 sr2 g2 1meg 
Rs3 g1 9 1meg 
Rs4 g2 10 1meg 
 
Vsr1 sr1 0 pulse(0.775 1.225 1n 1n 1n 1u 2u) 
Vsr2 sr2 0 pulse(1.225 0.775 1n 1n 1n 1u 2u) 
 
.tran 0 4u 
.control 
Run 
 
plot v(sr1) v(10,9) 
plot deriv(v(10,9)) 
.endc 
.end 



 
 

 Section 8:  

Transient simulation for maximum output voltage swing in closed 

loop mode. 

 

 



 

Partial Netlist  for Output Swing 

Rs1 sr1 g1 1meg 
Rs2 sr2 g2 1meg 
Rs3 g1 9 1meg 
Rs4 g2 10 1meg 
 
Vsr1 sr1 0 sin(0.93 0.4 1k 0 0) 
Vsr2 sr2 0 sin(0.93 -0.4 1k 0 0) 
.tran 0 4m 
.control 
run 
 
plot v(sr1) v(10,9) 
 
.endc 
.end 

 

 

From simulation we see that closed loop unity gain is achieved along 

with the swing of 1.6V 



 Section 9:  

 

 

Common-mode (CM) DC simulation 

 

 

 

From the above graph, it can be inferred that the output is stable from 

0.7v to 1.4v and its variation is less with input voltage.   


